
1. Change: Top-level test cases
	Function name
	TC_13_5_1a, TC_19_2_1

	Reason for change
	system interface needs to be explicitly specified in functions doing the mapping of the system ports

	Summary of change
	system interface handed over to function doing the mapping of the system ports
NOTE: Below the two test cases are chosen as examples for the different configurations used within test suite LTE_A_R12

	TTCN module
	LTE_A_R12_Testsuite.ttcn


	module LTE_A_R12_Testsuite {
...

  testcase TC_13_5_1a() runs on MTC_LTE system SYSTEM_LTE {

    // @purpose

    //   MTSI MO speech call / SSAC in Connected mode / 0% access probability for MTSI MO speech call

    var EUTRA_PTC        v_EUTRA      := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(360);

    v_EUTRA := EUTRA_PTC.create alive;

    v_IMS1 := IMS_PTC.create alive;

    f_MTC_ConnectPTCs_LTE(system, v_EUTRA, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_13_5_1a_EUTRA());

    v_IMS1.start(f_TC_13_5_1a_IMS1());

    t_GuardTimer.start;

    f_MTC_MainLoop_LTE(t_GuardTimer);

  }
...

  testcase TC_19_2_1() runs on MTC_LTE_D2D system SYSTEM_LTE_D2D {

    // @purpose

    //   ProSe Direct Discovery Monitoring/Pre-configured authorisation / Monitoring / Handling of validity timers / Utilisation of the resources of different cells/PLMNs

    var EUTRA_D2D_PTC    v_EUTRA      := null;

    var ProSe_PTC        v_ProSe      := null;

    var SideLink_PTC     v_SideLink   := null;

    timer t_GuardTimer := int2float(360);

    v_EUTRA := EUTRA_D2D_PTC.create alive;

    v_ProSe := ProSe_PTC.create alive;

    v_SideLink := SideLink_PTC.create alive;

    f_MTC_ConnectPTCs_LTE_D2D(system, v_EUTRA, v_ProSe, v_SideLink);

    v_EUTRA.start(f_TC_19_2_1_EUTRA());

    v_ProSe.start(f_TC_19_2_1_ProSe());

    v_SideLink.start(f_TC_19_2_1_SideLink());

    t_GuardTimer.start;

    f_MTC_MainLoop_LTE_D2D(t_GuardTimer);

  }
...

}


2. Change: f_MTC_ConnectPTCs_*

	Function name
	f_MTC_ConnectPTCs_LTE, f_MTC_ConnectPTCs_LTE_D2D

	Reason for change
	system interface explicitly handed over from top-level test case functions

	Summary of change
	explicit handling of system interface

	TTCN module
	MTC_Main_LTE, MTC_Main_LTE_D2D


	  function f_MTC_ConnectPTCs_LTE(SYSTEM_LTE p_System,

                                 EUTRA_PTC p_Eutra,

                                 IMS_PTC p_ImsPdn1,

                                 IMS_PTC p_ImsPdn2) runs on MTC_LTE

  { /* @sic R5-104796 change 12: CDMA2000 added sic@ */

    /* @sic R5-113805, R5-113696, R5-113734: new parameter p_RatCombination sic@ */

    /* @sic R5-113037: Introduction of positioning test cases sic@ */

    /* p_RatCombination is used only in cases where a dummy PTC is used for GERAN or UTRAN to get the system info for EUTRA's system information combination 10 and c10a */

    /* @sic R5-113037: Introduction of positioning test cases sic@ */

    var IP_PTC v_IP_PTC;

    var NASEMU_PTC v_NASEMU_PTC;

    var IMS_PTC v_ImsPdn1;

    var boolean v_C2KTunnellingFlag := false;

    vc_Components.Eutra := p_Eutra;

    if (p_Eutra == null) {

      FatalError (__FILE__, __LINE__, "invalid configuration");

      mtc.stop;

    }

    // ***** MTC *****

    f_MTC_BASE_Map(p_System);

    // ***** EUTRA PTC *****

    f_EUTRA_PTC_Map(p_System, p_Eutra, v_C2KTunnellingFlag);              // map EUTRA system ports

    v_NASEMU_PTC := f_NasEmu_CreateConnectAndMap(p_System, p_Eutra);      // create NasEmu, map system ports and connect with EUTRA PTC

    v_NASEMU_PTC.start(f_NASEMU_MainLoop());                    // start NASEMU

    // ***** Create, map, connect and start IP PTC *****

    v_IP_PTC := f_IP_PTC_CreateAndMap(p_System);

    if (p_Eutra != null)     {connect(p_Eutra:IP, v_IP_PTC:EUTRA_CTRL);}    /* @sic R5-125676 sic@ */

    v_IP_PTC.start(f_IP_PTC_MainLoop());

    // ***** Create, connect and start IMS PTC *****

    // IMS for PDN1: if PTC is not created yet the default IMS PTC will be used

    if (p_ImsPdn1 == null) { v_ImsPdn1 := f_IMS_PTC_Create(); }         // => default IMS handling for PDN1

    else                   { v_ImsPdn1 := p_ImsPdn1; }

    connect(v_ImsPdn1:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN1]);

    connect(v_ImsPdn1:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN1]);

    connect(v_ImsPdn1:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN1]);

    connect(v_ImsPdn1:IPCAN,         p_Eutra:IMS[tsc_Index_PDN1]);

    connect(v_ImsPdn1:MMI,           mtc:PTC_Ut[tsc_MTC_PortIndex_IMS1]);   /* @sic R5-134070: generic handling of UT commands for IMS PTC as well as for the other PTCs sic@ */

    if (p_ImsPdn1 == null) {

      v_ImsPdn1.start(f_IMS_DefaultHandler());

    }

    // IMS for PDN (emergency call)

    if (p_ImsPdn2 != null) {                             // test case specific IMS handling for PDN2

      connect(p_ImsPdn2:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN2]);

      connect(p_ImsPdn2:IPCAN,         p_Eutra:IMS[tsc_Index_PDN2]);

      connect(p_ImsPdn2:MMI,           mtc:PTC_Ut[tsc_MTC_PortIndex_IMS2]);   /* @sic R5-134070: generic handling of UT commands for IMS PTC as well as for the other PTCs sic@

                                                                                 @sic R5s130987 sic@ */

    }

    // ***** Connect PTCs to MTC *****

    if (p_Eutra != null)     {connect(mtc:PTC_Ut[tsc_MTC_PortIndex_EUTRA],    p_Eutra:UT);}

    f_UT_InitialPowerOffUE(Ut); /* @sic R5s120520 sic@ */

  }

	  function f_MTC_ConnectPTCs_LTE_D2D(SYSTEM_LTE_D2D p_System,

                                     EUTRA_D2D_PTC p_Eutra,

                                     ProSe_PTC p_ProSe,

                                     SideLink_PTC p_SideLink) runs on MTC_LTE_D2D

  {

    var IP_PTC v_IP_PTC;

    var IMS_PTC v_IMS_PTC;

    var HTTP_ProSe_PTC v_HTTP_PTC;

    var NASEMU_PTC v_NASEMU_PTC;

    var boolean v_DnsRequired := true;

    var boolean v_HttpRequired := true;

    var boolean v_UdpClientRequired := true;

    vc_Components.Eutra := p_Eutra;

    vc_Components.ProSe := p_ProSe;

    vc_Components.SideLink := p_SideLink;

    if ((p_Eutra == null) or ((p_ProSe == null) and (p_SideLink == null))) {

      FatalError (__FILE__, __LINE__, "invalid configuration");

      mtc.stop;

    }

    // ***** MTC *****

    f_MTC_BASE_Map(p_System);

    // ***** EUTRA PTC *****

    f_EUTRA_PTC_Map(p_System, p_Eutra);                                   // map EUTRA system ports

    v_NASEMU_PTC := f_NasEmu_CreateConnectAndMap(p_System, p_Eutra);      // create NasEmu, map system ports and connect with EUTRA PTC

    v_NASEMU_PTC.start(f_NASEMU_MainLoop());                    // start NASEMU

    v_IP_PTC := f_IP_PTC_CreateAndMap(p_System);

    connect(p_Eutra:IP, v_IP_PTC:EUTRA_CTRL);

    v_IP_PTC.start(f_IP_PTC_MainLoop(v_DnsRequired, v_HttpRequired, v_UdpClientRequired));

    // ***** Create, connect and start IMS PTC *****

    v_IMS_PTC := f_IMS_PTC_Create();

    connect(v_IMS_PTC:IMS_CTRL,      v_IP_PTC:IMS_CTRL[tsc_Index_PDN1]);

    connect(v_IMS_PTC:IMS_Server,    v_IP_PTC:IMS_Server[tsc_Index_PDN1]);

    connect(v_IMS_PTC:IMS_Client,    v_IP_PTC:IMS_Client[tsc_Index_PDN1]);

    connect(v_IMS_PTC:IPCAN,         p_Eutra:IMS[tsc_Index_PDN1]);

    v_IMS_PTC.start(f_IMS_DefaultHandler());

    // ***** Connect EUTRA to MTC *****

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_EUTRA], p_Eutra:UT);

    // ***** Connect PROSE *****

    if (p_ProSe != null) {

      v_HTTP_PTC := f_HTTP_ProSe_PTC_Create();

      connect(p_ProSe:PROSECTRL,  v_HTTP_PTC:PROSECTRL);

      connect(p_ProSe:PROSEDATA,  v_HTTP_PTC:PROSEDATA);

      connect(v_IP_PTC:UDPCLIENT, v_HTTP_PTC:OMAPUSH);

      connect(v_IP_PTC:HTTPCTRL,  v_HTTP_PTC:HTTPCTRL);

      connect(v_IP_PTC:HTTPDATA,  v_HTTP_PTC:HTTPDATA);

      v_HTTP_PTC.start(f_HTTP_ProSe_MainLoop());

      connect(p_ProSe:EUTRA,     p_Eutra:PROSE);

    }

    // ***** Connect SIDELINK *****

    if (p_SideLink != null) {

      connect(p_SideLink:EUTRA,  p_Eutra:SIDELINK);

      f_SideLink_PTC_Map(p_System, p_SideLink);

    }

    f_UT_InitialPowerOffUE(Ut);

  }


3. Change: Functions mapping the system ports
	Function name
	f_MTC_BASE_Map, f_EUTRA_PTC_Map, f_NasEmu_CreateConnectAndMap, f_IP_PTC_CreateAndMap, f_SideLink_PTC_Map

	Reason for change
	system interface explicitly handed over from top-level test case functions

	Summary of change
	replacement of "system" by handed over system interface

	TTCN module
	MTC_Base, EUTRA_Component, NasEmu_Component, IP_PTC_Component, f_SideLink_PTC_Map


	  function f_MTC_BASE_Map(MTC_BASE_SYSTEM p_System)

  {

    map(mtc:Ut, p_System:Ut);

  }

	  function f_EUTRA_PTC_Map(EUTRA_SYSTEM p_System,

                           EUTRA_PTC p_Eutra,

                           boolean p_C2KTunneling := false)

  {

    map(p_Eutra:SYS,    p_System:E_SYS);

    map(p_Eutra:SYSIND, p_System:E_SYSIND);

    map(p_Eutra:DRB,    p_System:E_DRB);

    if (f_GetTestcaseAttrib_Qbased_Rsrq(testcasename())) {    /* virtual noise generator for RSRQ test cases */

      map(p_Eutra:VNG,  p_System:E_VNG);

    }

    if (f_GetTestcaseAttrib_Mbms(testcasename())) {    /* Map the MRB port only for MBMS test cases */

      map(p_Eutra:MRB,  p_System:E_MRB);

    }

    if (p_C2KTunneling) {

      map(p_Eutra:C2KTUNNEL, p_System:E_C2KTUNNEL);

    }

  }

	  function f_NasEmu_CreateConnectAndMap(NASEMU_SYSTEM p_System,

                                        EUTRA_PTC p_Eutra) return NASEMU_PTC

  {

    var NASEMU_PTC v_NASEMU_PTC := null;

    if (p_Eutra != null) {

      // create NAS emulation:

      v_NASEMU_PTC := NASEMU_PTC.create alive;

      // connect NAS emulation:

      connect(p_Eutra:NASCTRL,  v_NASEMU_PTC:CTRL);

      connect(p_Eutra:SRB,      v_NASEMU_PTC:TC_SRB);

      // map NASEMU system ports:

      map(v_NASEMU_PTC:SYS_SRB, p_System:E_SRB);

    }

    return v_NASEMU_PTC;

  }

	  function f_IP_PTC_CreateAndMap(IP_SYSTEM p_System) return IP_PTC

  {

    var IP_PTC v_IP_PTC := IP_PTC.create alive;

    var IPMUX_PTC v_IPMUX_PTC := IPMUX_PTC.create alive;

    map(v_IP_PTC:IP_CTRL,    p_System:IP_CTRL);

    map(v_IP_PTC:IPSEC_CTRL, p_System:IPSEC_CTRL);

    map(v_IPMUX_PTC:IP_SOCK, p_System:IP_SOCK);

    connect(v_IP_PTC:IP_SOCK_DATA, v_IPMUX_PTC:IP_SOCK_DATA);

    connect(v_IP_PTC:IP_SOCK_CTRL, v_IPMUX_PTC:IP_SOCK_CTRL);

    v_IPMUX_PTC.start(f_IPMUX_PTC_MainLoop());

    return v_IP_PTC;

  }

	  function f_SideLink_PTC_Map(SIDELINK_SYSTEM p_System,

                              SideLink_PTC p_SideLink)

  {

    map(p_SideLink:SYS,    p_System:SL_SYS);

    map(p_SideLink:SYSIND, p_System:SL_SYSIND);

    map(p_SideLink:DATA,   p_System:SL_DATA);

  }


